
1. Introducción

Para algunas funciones de C-Gate es necesario que la información se dé a C-Gate desde el programa Carola. Consulte en los
respectivos capítulos los bloques necesarios y las instrucciones para la funcionalidad correspondiente.

Si utiliza la estructura de datos ReadVar o el comando OPCUAVAR en otro contexto, asegúrese de que las áreas utili-
zadas no se superpongan. Tenga también en cuenta que el índice utilizado en el programa Carola no es el mismo índice en
la interfaz web. Los valores se pasan a C-Gate a través de la estructura de datos ReadVar . En el controlador, el primer
índice es uno y el último 100. En C-Gate, el primer índice es 0 y el último 99. Un valor que se almacena en el índice 50 en el
controlador puede recuperarse en el índice 49 en C-Gate.

Tenga en cuenta que los ejemplos del Código Carola que se explican a continuación pueden estar ya parcial o totalmente
almacenados en el programa del Máster. Por favor, compare las secciones y adáptelas si es necesario. La duplicación de
funciones y variables es reconocida como un error por el intérprete Carola.

Nota: Los cambios en las variables de QIROX WriteVar pueden afectar al flujo del programa Carola.

Nota: Por favor, utilice Adobe Acrobat Reader para abrir este archivo si quiere copiar el código del programa Carola.

2. Eficacia global de los equipos (OEE)

2.1. Determinar la cantidad real y la cantidad buena

Para determinar la cantidad real y la cantidad buena, deben almacenarse las siguientes secciones del programa en el pro-
gramamaestro y en el programa de componentes, respectivamente. Es importante que los siguientes procedimientos uti-
licen exclusivamente el rango de índices de 94 a 100 y no puedan utilizarse para otros fines. La gama puede ampliarse en el
futuro.

2.1.1. Programa de Máster

Las primeras líneas son necesarias para que las variables puedan ser importadas y escritas desde otros programas. Se en-
cuentran directamente después de la instrucción RESTART .

RESTART
! DECLARACIÓN DE VARIABLES
VAR SEL, ERG
STRING VAR COMPSTAT, COMPQUAL
PUBLIC STRING VAR COMPTYPE,COMPVARI,COMPSERI,COMPSTEP,COMPSEAM

Este bloque define un procedimiento que pasa a C-Gate la información de que un componente se está iniciando. Previa-
mente, las variables que se utilizan a continuación deben estar configuradas en el programa del componente.

PUBLIC PROC COMPSTART
COMPSTAT := ''
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
WAITM(100)

STRHND(2, COMPTYPE, COMPTYPE, ERG)
IF ERG=0 THEN COMPTYPE:='NODATA'
STRHND (0,COMPTYPE,'NODATA',ERG)
IF ERG=1 THEN BEGIN
LOPCUA1:
WRITE ('VARIABLE >COMPTYPE< AÚN NO DESCRITA!')
WAITM(500)
PAUSE
JUMP LOPCUA1
END

COMPSTAT := 'START'
COMPQUAL := 'IN_PROGRESS'
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99, COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98, COMPSERI,'COMPSERNO')
OPCUAVAR(0,97, COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95, COMPQUAL,'COMPQUALITY')
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
ENDP

Este bloque define un procedimiento que pasa a C-Gate la información de que un componente ha terminado.

! FIN DEL REGISTRO DE COMPONENTES
PUBLIC PROC COMPEND
COMPSEAM := ''
COMPSTAT := 'END'
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
WAITM(100)

COMPTYPE := ''
COMPVARI := ''
COMPSERI := ''
COMPSTEP := ''
COMPSTAT := ''
COMPQUAL := ''
COMPSEAM := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99, COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98, COMPSERI,'COMPSERNO')
OPCUAVAR(0,97, COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96, COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95, COMPQUAL,'COMPQUALITY')

OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
WAITM(100)
ENDP

Este bloque define una función que pasa a C-Gate la información de que se está iniciando una nueva costura. El nombre de
la costura debe escribirse en la variable COMPSEAM antes de llamar al procedimiento.

! INICIAR UNA COSTURA
PUBLIC PROC SEAMSTRT
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
WAITM(100)
ENDP

Este bloque define una función que pasa a C-Gate la información de que la costura actual ha terminado.

! FIN DE UNA COSTURA
PUBLIC PROC SEAMEND
COMPSEAM := ''
OPCUAVAR(0,94, COMPSEAM,'COMPSEAMNAME')
WAITM(100)
ENDP

Este bloque define una función que consulta en el PHG si el componente está bien o no. El control se basa en una inspección
visual por parte del trabajador y bloquea la secuencia posterior del programa hasta que se haya realizado una entrada.
Puede adaptar la redacción ¿EL COMPONENTE ESTÁ BIEN=1 O NO ESTÁ BIEN=0? a sus necesidades.

! GRABACIÓN DE COMPONENTES DE CALIDAD
PUBLIC PROC COMPQMAN
WREAD ('¿EL COMPONENTE ESTÁ BIEN=1 O NO ESTÁ BIEN=0?', SEL )
IF SEL=1 THEN COMPQUAL := 'OK' ELSE COMPQUAL := 'NOT_OK'
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
WAITM(100)
ENDP

Este bloque define una función que automáticamente (sin consulta) establece la calidad del componente en OK. Esta fun-
ción se puede utilizar si no se desea registrar la calidad de la eficacia global de los equipos (OEE) de forma dinámica, pero
se desea que se calcule correctamente toda la métrica.

! GRABACIÓN DE COMPONENTES DE CALIDAD
PUBLIC PROC COMPQAUT
COMPQUAL := 'OK'
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
WAITM(100)
ENDP

2.1.2. Variables

En el programa se pueden utilizar las siguientes variables.

COMPTYPE : Tipo de componente.

COMPVARI : Variante del componente.

COMPSERI : Número de serie del componente. Si no se establece ningún valor, C-Gate crea automáticamente un
número de serie basado en unamarca de tiempo.

COMPSTEP : Nombre de la etapa de producción. El valor es importante si el componente se procesa en varias esta-
ciones.

COMPSEAM : Nombre de la costura. Una costura puede estar formada por varios segmentos de la web.

2.1.3. Programa de componentes simples

Importe las variables y los procedimientos del programamaestro.

RESTART

! IMPORTACIÓN DE VARIABLES Y PROCEDIMIENTOS
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSERI,COMPSTEP,COMPSEAM FROM MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER

MAIN

En el flujo del programa, establezca al menos los campos obligatorios y, si es necesario, también los opcionales. En cuanto
todos los camposesténconfigurados, llamealprocedimiento COMPSTART . Si haadquiridoelmóduloC-Gate.QM, también
puede definir costuras y estructurar el protocolo de los componentes de forma más precisa. Para ello, utilice los procedi-
mientos SEAMSTRT y SEAMEND . En cuanto el componente esté terminado y quiera ajustar la calidad del componente
desde el programa Carola, llame a COMPQAUT o a COMPQMAN para transferir los datos de calidad. Si ha adquirido el
módulo C-Gate.QM, la calidad de la pieza se determina automáticamente mediante la supervisión QIROX SD. Llamar al
procedimiento COMPQAUT o COMPQMAN no tiene ningún efecto sobre el protocolo del componente.

Para terminar el componente, llame al procedimiento COMPEND . El orden de las llamadas es importante.

Tenga en cuenta que debe transcurrir hasta un segundo entre la llamada a COMPEND y el reinicio de un componente con
COMPSTART , de lo contrario la señal de fin puede perderse debido a una sincronización de datos en el controlador. Está
previsto que esta limitación se elimine en una de las próximas versiones de software para el controlador del robot.

! CONFIGURACIÓN DE LA INFORMACIÓN DE LOS COMPONENTES
! CAMPO OBLIGATORIO
COMPTYPE := 'ARMARIO DE CONTROL'

! OPCIONAL:
! COMPVARI := 'XL'

! OPCIONAL:

! COMPSERI := '1234567890'

! OPCIONAL:
! COMPSTEP := 'PISO'

! INICIO DEL COMPONENTE
CALL COMPSTART

! ***************
! PROGRAMA DE COMPONENTES
! ***************

COMPSEAM := 'SOLDADURA EN ÁNGULO 1'
CALL SEAMSTRT

! ***************
! PROGRAMA SOLDADURA EN ÁNGULO 1. POR EJEMPLO, VARIOS MOVIMIENTOS DE LA GC.
! ***************

COMPSEAM := 'SOLDADURA EN ÁNGULO 2'
CALL SEAMSTRT

! ***************
! PROGRAMA SOLDADURA EN ÁNGULO 2. POR EJEMPLO, VARIOS MOVIMIENTOS DE LA GC.
! ***************

CALL SEAMEND

! ESTABLECER LA CALIDAD DEL COMPONENTE EN EL PROGRAMA CAROLA
! LLAMAR ANTES DE COMPEND
CALL COMPQMAN
! O
! CALL COMPQAUT

! FIN DEL COMPONENTE
! LLAME A COMPQMAN O COMPQAUT
CALL COMPEND

! FIN DEL PROGRAMA DE COMPONENTES
END

2.1.4. Programa de componentes complejos

El siguienteprogramadeCarola creauncomponente condospasosdeproducción. Lospasosdeproducciónpuedensoldar-
se de forma secuencial en un robot o en paralelo en varios robots. El requisito previo para ello es que los pasos se combinen

en un protocolo:

El tipo, la variante y el número de serie deben ser iguales en todos los robots.

El modelo de componente correspondiente debe tener el número mínimo correcto de pasos de producción. En este
caso, se sueldan dos pasos de producción, por lo que el número debe fijarse comomínimo en el valor dos.

RESTART
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSERI,COMPSTEP,COMPSEAM FROM MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTRT,SEAMEND FROM MASTER
MAIN

! ***************
! FASE DE PRODUCCIÓN 1
! ***************

COMPTYPE := 'CALDERA'
COMPVARI := '80KW'
COMPSERI := '0042012342'
COMPSTEP := 'PISO'

CALL COMPSTART

COMPSEAM := 'SOLDADURA EN ÁNGULO 1'
CALL SEAMSTRT
! PROGRAMA SOLDADURA EN ÁNGULO 1. POR EJEMPLO, VARIOS MOVIMIENTOS DE LA GC.

COMPSEAM := 'SOLDADURA EN ÁNGULO 2'
CALL SEAMSTRT
! PROGRAMA SOLDADURA EN ÁNGULO 2. POR EJEMPLO, VARIOS MOVIMIENTOS DE LA GC.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

! ***************
! FASE DE PRODUCCIÓN 2
! ***************

COMPTYPE := 'CALDERA'
COMPVARI := '80KW'
COMPSERI := '0042012342'
COMPSTEP := 'PORTADA'

CALL COMPSTART

COMPSEAM := 'SOLDADURA EN ÁNGULO 3'
CALL SEAMSTRT
! PROGRAMA SOLDADURA EN ÁNGULO 3. POR EJEMPLO, VARIOS MOVIMIENTOS DE LA GC.

COMPSEAM := 'SOLDADURA EN ÁNGULO 4'
CALL SEAMSTRT
! PROGRAMA SOLDADURA EN ÁNGULO 4. POR EJEMPLO, VARIOS MOVIMIENTOS DE LA GC.

CALL SEAMEND
CALL COMPQAUT
CALL COMPEND

END

3. Notas

Si usted es o ha sido cliente de PDMoUMS y una de las dos opciones está activada en la configuración del controlador, ¡hay
que observar urgentemente la siguiente nota! Si se activa una de las dos opciones, pero el controlador del robot no está
conectado a la aplicación dePC correspondiente, se producirán lagunas en el flujo de datos que pueden llegar a durar hasta
5 segundos y falsear de forma significativa el protocolo del componente. Conecte el controlador del robot a la aplicación
del PC o elimine la opción correspondiente de la configuración.

Si está trabajando con varios robots al mismo tiempo en el mismo componente y desea que todos los datos se fusionen en
un protocolo de componente en el marco del módulo C-Gate.QM, póngase en contacto con el Servicio CLOOS para obtener
más información. En caso contrario, deje el campo COMPSERI vacío, ya que esto puede provocar efectos secundarios no
deseados en el cálculo de la eficacia global del equipo (OEE).

4. Variables

Tiene la posibilidad de adjuntar variables del programa Carola al protocolo del componente. Para ello, utilice el comando
OPCUAVAR (ver ejemplo). Si el nombre (último parámetro) tiene el prefijo ’COMPVAR_’, la variable se adjunta al compo-
nente. Si el nombre tiene el prefijo ’STEPVAR_’, la variable se adjunta al paso del componente. Si el nombre tiene el prefijo
’SEAMVAR_’, la variable se adjunta a la costura actual. Sólo se puede almacenar un valor para cada nombre seleccionado. El
último valor que se estableció con el nombre respectivo se almacena en el registro. El prefijo del nombre no se almacena. El
valor debe restablecerse antes de que se complete la costura, el paso o el componente. Si no se hace así, el valor se añade
también al siguiente elemento. Se pueden guardar números enteros, comas y textos.

CALL SEAMSTRT
OPCUAVAR(0,1, 'EXAMPLE-VALUE','COMPVAR_SEAM_MEASURE_A')

OPCUAVAR(0,1, '','')
CALL SEAMEND

